
C1M12

Extrema

A while back in C1M10 we discussed the Mean Value Theorem, but only lightly. We used it to
determine that a continuous function must be increasing on an interval where the derivative is positive,.
Similarly, a negative derivative signals a decreasing function. Let’s recall the two hypotheses of the MVT
and then add a third.
H1: f is a continuous function on the closed interval [a, b] .
H2: f is differentiable on the open interval (a, b).
H3: There is a point c in (a, b) for which f(x) ≤ f(c) for all x in some open interval containing c .

The conclusion will be stated below. We are going to approach c from each side and compare the
results.

Assume x < c . Then
f(x) − f(c)

x − c
≥ 0, so lim

x→c−

f(x) − f(c)
x − c

= f ′(c) ≥ 0.

Assume x > c . Then
f(x) − f(c)

x − c
≤ 0, so lim

x→c+

f(x) − f(c)
x − c

= f ′(c) ≤ 0.

Taken together, f ′(c) ≥ 0 and f ′(c) ≤ 0 imply f ′(c) = 0. The one-sided limits existed because f ′(c)
existed. When there is an open interval containing a point and that point satisfies f(x) ≤ f(c) for all x in
the interval, we say that f has a local maximum or relative maximum at c .

In a similar manner, suppose that we replace H3 with
H4: There is a point c in (a, b) for which f(x) ≥ f(c) for all x in some open interval containing c .

In this case f has a local minimum or relative minimum at c . By an obvious parallel argument we have
the same conclusion. So, with either H3 or H4 our conclusion is :
C: f ′(c) = 0.

This tells us that when our function is differentiable the only way we can have a local extreme (local
maximum or local minimum) that is not an endpoint, is for the derivative to be zero at the point. For this
reason we define a point c to be a critical point if either the derivative does not exist at c or f ′(c) = 0. The
Extreme Value Theorem was mentioned in the module on continuity. From it we learned that a continuous
function on a closed interval assumes a maximum value and a minimum value.

Here we have the graphs of two continuous functions on [0, 4], f and g . We have labeled the extrema
of f , except for the endpoints which are a local minimum (x = 0) and a local maximum (x = 4). It is clear

that at x = 1 the function g has a maximum, but the derivative does not exist there. The endpoints serve
as a local minimum (x = 0) and a minimum (absolute) (x = 4) for g .
Maple Example: The function f in the diagram above is actually defined by f(x) = x sin(πx). Let’s
use Maple to locate all extrema of f . We will show all steps, including the plotting of text. After finding our
first relative maxima, at c1 , we will test the values of f on both sides of c1 . We will also test the derivative
of f on both sides.
> restart: with(plots): with(plottools):
> f:=x->x*sin(Pi*x); a:=0; b:=4;

f := x → x sin(πx)
a := 0
b := 4

> fprime:=unapply(diff(f(x),x),x);
fprime := x → sin(πx) + x cos(πx)π

This command takes the derivative of f(x) with respect to x and sets it up as a function.
> f2prime:=unapply(diff(f(x),x,x),x);

f2prime := x → 2 cos(πx)π − x sin(πx)π2

Note how this command takes the second derivative of f(x) and sets it up as a function.
> c1:=fsolve(fprime(x)=0,x=0..1);

c1 := .6457736765
> m1:=evalf(f(c1));

m1 := .5792303274
> evalf(f(c1-.02)); evalf(f(c1+.02));

.5775557266

.5775085314
> evalf(fprime(c1-.02)); evalf(fprime(c1+.02));

.1662069815
−.1732844003

From the last few lines we see that the derivative of f is positive on the left of c1 and negative on the right
of c1 . This means that as we move from left to right we are going uphill, level, and then downhill − a sure
sign that c1 yields a relative maximum. Later we will state all this as a theorem.
> evalf(f2prime(c1));

−8.494699804
This is overkill, but we may also use the second derivative to see if the first derivative is a decreasing function,
and the negative value of f2prime at c1 says that it is, thus signalling that f is concave down at c1 and
there is a relative maximum there.
> c2:=fsolve(fprime(x)=0,x=1..2);

c2 := 1.563913906
> m2:=evalf(f(c2));

m2 := −1.532493362
> evalf(f2prime(c2));

16.37825282 Concave up, relative minimum
> c3:=fsolve(fprime(x)=0,x=2..3);

c3 := 2.539688175
> m3:=evalf(f(c3));

m3 := 2.519972588
> evalf(f2prime(c3));

−25.65251748 Concave down, relative maximum
> c4:=fsolve(fprime(x)=0,x=3..4);

c4 := 3.528636468
> m4:=evalf(f(c4));

m4 := −3.514366510
> evalf(f2prime(c4));

35.24990626 Concave up, relative minimum
> A:=f(a); B:=f(b);

A := 0

B := 0
> M:=max(m1,m2,m3,m4,A,B);

M := 2.519972588
> m:=min(m1,m2,m3,m4,A,B);

m := −3.514366510
> B1:=plot(f(x),x=0..4,color=blue,scaling=constrained):
> B2:=line([c1,0],[c1,f(c1)],color=red):
> B3:=line([c2,0],[c2,f(c2)],color=red):
> B4:=line([c3,0],[c3,f(c3)],color=red):
> B5:=line([c4,0],[c4,f(c4)],color=red):
> B21:=textplot([c1,-.3,"c1"]):
> B31:=textplot([c2,.3,"c2"]):
> B41:=textplot([c3,-.3,"c3"]):
> B51:=textplot([c4,.3,"c4"]):
> B6:=plot(m,x=0..4,color=green):
> B61:=textplot([2,m+.3,"y=m"]):
> B7:=plot(M,x=0..4,color=magenta):
> B71:=textplot([1.6,M-.24,"y=M"]):
> B8:=textplot([3.35,1.5,"y=f(x)"]):
> display(B1,B2,B3,B4,B5,B21,B31,B41,B51,B6,B61,B7,B71,B8);

Maple Example: Find the minimum distance from Q(3/2, 1) to the graph of f(x) = x sin(πx).

Frequently calculus texts include an exercise to find the minimum distance from a given point to a
simple curve, such as a parabola. This is a similar, yet more challenging, problem involving the function in
the previous example. We will exploit the power of Maple and have it find numerical solutions for critical
points that we would be unlikely to locate easily. The function which determines the distance from the point
Q(3/2, 1) to the point (x, f(x)) on the curve is designated as R(x), its derivative is S(x), and its second
derivative is T (x). Because T (x) is a bit cumbersome, its display is suppressed with a colon. Since several
points may occur as solutions, we will use the Second Derivative Test to determine whether a critical point
is a relative maximum or minimum. The derivative has a denominator that looks as if it could be a problem.
However, a close examination reveals that it is simply the distance function R(x). Since Q does not lie on
the curve, the denominator won’t be 0, but we will check it anyway. We chose to make the derivatives of f
functions instead of expressions.

> restart: with(plots): with(plottools):

> f:=x->x*sin(Pi*x);
f := x → x sin(πx)

> Q:=[3/2,1];

Q :=
[

3
2
, 1

]

Set up our distance formula.
> R:=x->sqrt((x-Q[1])ˆ2+(f(x)-Q[2])ˆ2);

R := x →
√

(x − Q1)2 + (f(x) − Q2)2

Set up the first derivative as a function.
> S:=unapply(diff(R(x),x),x);

S := x → 1
2

(2x − 3 + 2(x sin(πx) − 1)(sin(πx) + x cos(πx) π))√
(x − 3/2)2 + (x sin(πx) − 1)2)

Let’s take a look at that denominator. If it has a zero in the interval, then we have a problem. We will
“eyeball” it rather than solving the equation.
> De:=denom(S(x));

2(
√

(x − 3/2)2 + (x sin(πx) − 1)2)
> D1:=plot(De,x=0..3.2,color=red):
> D2:=plot(S(x),x=0..3.2,color=blue):
> D3:=textplot([1.55,6,"denominator of derivative"]):
> D4:=textplot([1.5,-3.5,"derivative"]):
> display(D1,D2,D3,D4);

Set up the second derivative of the function as a function.
> T:=unapply(diff(R(x),x,x),x):
> P0:=plot(f(x),x=0..3.2,color=blue):
> p1:=fsolve(S(x)=0,x,1/2..5/2);

p1 := 1.565491407
> evalf(T(p1));

−16.00806205
T (p1) = R′′(p1) < 0 ⇒ relative maximum
> P1:=pointplot([p1,f(p1)],symbol=circle):
> p2:=fsolve(S(x)=0,x,1.6..5/2);

p2 := 2.138585926
> evalf(T(p2));

67.70258030

4



T (p2) = R′′(p2) > 0 ⇒ relative minimum
> P2:=pointplot([p2,f(p2)],symbol=circle):
> p3:=fsolve(S(x)=0,x,.5..1.4);

p3 := .7918557211
> evalf(T(p3));

9.019878695
T (p3) = R′′(p3) > 0 ⇒ relative minimum
> P3:=pointplot([p3,f(p3)],symbol=circle):
> A:={evalf(R(p1)),evalf(R(p2)),evalf(R(p3))};

A := {2.533319655, .6460679294, .8775682312}
> mindistance:=min(op(A));

mindistance := .6460679294
We used op(A) in order to access the components of the set A . We have our answer and the point at which
it occurs, (p2, f(p2)). Now we will display a graph that illustrates the items which we found.
> P4:=pointplot(Q,symbol=diamond,symbolsize=15):
> P5:=line(Q,[p1,f(p1)],color=green):
> P6:=line(Q,[p2,f(p2)],color=green):
> P7:=line([p2,f(p2)],[p2,0],color=magenta):
> P8:=line(Q,[p3,f(p3)],color=green):
> P9:=plot(R(x),x=0..3.3,color=red):
> T1:=textplot([2.8,-1.7,"y=f(x)"]):
> T2:=textplot([2.8,3.7,"R(x)=distance"]):
> T3:=textplot([1.5,1.4,"Q"]):
> T4:=textplot([p2,-.35,"p2"]):
> display(P0,P1,P2,P3,P4,P5,P6,P7,P8,P9,T1,T2,T3,T4);

Procedure to Locate Extrema: It is time to summarize our approach to extrema. Suppose that we
have a continuous function f on a closed interval. We locate all points at which the first derivative of f
is zero or for which the derivative does not exist, calling such points critical points. We evaluate f at the
critical points and the endpoints and know that our global maximum and global minimum will be found
among these values. To determine whether a critical point for which f is differentiable yields a relative
maxima or minima, we apply the First Derivative Test.
First Derivative Test: Assume that f is differentiable on an open interval containing c and that
f ′(c) = 0.
(a) If the derivative f ′(x) changes sign from positive to negative as x moves through c from left to right,
then f has a relative maximum at c .

(a) If the derivative f ′(x) changes sign from negative to positive as x moves through c from left to right,
then f has a relative minimum at c .

Noting that in case (a) the derivative is decreasing and that in case (b) the derivative is increasing, we
state the Second Derivative Test for determining the status of critical points where the derivative is zero.
Second Derivative Test: Assume that f is differentiable on an open interval containing c and that
f ′(c) = 0. Assume that the second derivative, f ′′ , also exists on the open interval.
(a) If f ′′(c) < 0, then f has a relative maximum at c .
(b) If f ′′(c) > 0, then f has a relative minimum at c .

There is another concept to consider that we have not yet mentioned. What happens when f ′′(c) = 0?
The answer is inconclusive without additional information. A point of inflection occurs when a function
stops being concave one way and starts being concave the other way. This happens when the derivative
stops increasing and starts decreasing, or vice versa.
Point of Inflection: Suppose that f ′′(x) exists on an open interval containing c and that f ′′(c) = 0.
Then c is a point of inflection of f if f ′′(x) changes sign as x passes through c .
Examples: (a) If f(x) = x4 , then f ′(x) = 4x3 and f ′′(x) = 12x2 . If c = 0 then f ′′(c) = 0, but c is not
a point of inflection for f because f ′′ does not change sign as x passes through 0.
(b) Suppose that f ′′(x) = (x − 1)2(x − 2). Then f ′′ is zero at 1 and 2. But f ′′ changes sign only at 2, so
2 produces the only inflection point for f .
Maple Example: Find all points of inflection of f(x) = 3x5 − 10x4 + 10x3 − 60x2 + 5x + 2 and plot the
graph of f .
> restart: with(plots): with(plottools):
> f:=x->3*xˆ5-10*xˆ4+10*xˆ3-60*xˆ2+5*x+2;

f := x → 3x5 − 10x4 + 10x3 − 60x2 + 5x + 2
> f2prime:=unapply(diff(f(x),x,x),x);

f2prime := x → 60 x3 − 120 x2 + 60 x − 120
> factor(f2prime(x));

60 (x − 2) (x2 + 1)
> f2prime(1.98); f2prime(2.02);

−5.904480
6.096480

At 2, f stops being concave down and starts being concave up.
> U1:=plot(f(x),x=-1..4):
> U2:=line([2,0],[2,f(2)],color=blue):
> display(U1,U2);

–300

–200

–100

0

100

200

–1 1 2 3 4
x

6



C1M12 Problems:

1. Use Maple to find the critical points, points of inflection, and to display the graphs of the given functions,
their derivatives and second derivatives.
a. f(x) = x4 + 6x3 − 24x2 + 24 b. g(x) = 3x5 − 5x3

2. Given: f(x) = x cos(πx) on [1, 5].
a. Use Maple to find the relative maxima and minima and the global maximum and minimum as was done
in the Maple Example. Display vertical lines at the extrema as in the example.
b. Use Maple to locate the point P (x, y) on the graph of f(x) that is closest to the point Q(3, 5). Display
lines connecting the possible solutions to Q as was done in the Maple Example.

